



Creative Uses *for* **XMIM**

This practical guide offers useful examples for demonstrating how XMIM, LIM's patented near-English query tool, can help you in modeling and evaluating trading opportunities. The detailed studies within are designed to acquaint you with the software's capabilities and to highlight interesting and useful functionality. Each one demonstrates a problem, illustrates the query language and displays the results as would appear in the application.

Energy Applications

Volume 3: Queries and Results



LOGICAL INFORMATION MACHINES

Chicago: +1 (312) 456-3000
Houston: +1 (713) 652-4066
New York: +1 (212) 292-4282
Austin: +1 (512) 697-3000
London: +44 20 7190 2945
Singapore: +(65) 6322 4122

www.lim.com

www.markethistory.com

Copyright © 2006 by Logical Information Machines, Inc.
Patented May, 1995 U.S. Patent No. 08/392, 612
All rights reserved.

No part of this publication may be reproduced, stored in a retrieval system, or transmitted in any form or by any means, electronic, mechanical, photocopying, recording, or otherwise, without the prior written permission of Logical Information Machines, Inc.

Restricted Rights Legend

Use, duplication, or disclosure by the Government is subject to restrictions as set forth in subdivision (c) (1) (ii) of the Rights in Technical Data and Computer Software clause at 252.227-7013.

Logical Information Machines, Inc.
120 North LaSalle Street, Suite 2150
Chicago, IL 60602
+1 (312) 456-3000 (Direct)
+1 (312) 456-3475 (Fax)

Product names mentioned herein are for identification purposes only and may be trademarks and/or registered trademarks of their respective companies.

While every precaution has been taken in the preparation of this manual, we assume no responsibility for errors or omissions. Neither is any liability assumed for damages resulting from the use of the information contained herein. Logical Information Machines, Inc. may revise this publication from time to time without notice.

Table of Contents

Introduction	4
U.S. Petroleum Inventories & NYMEX WTI Price Declines	5
U.S. Petroleum Inventories & NYMEX WTI Prices	7
NYMEX WTI Trend Analysis	9
NYMEX WTI—Commitment of Traders & Small Traders	11
NYMEX WTI—Commitment of Traders Graph	13
NYMEX Natural Gas & Inventories	14
NYMEX Natural Gas & Summer Temperatures	16
NYMEX Natural Gas Prices & Hurricanes	18
Winter Season NYMEX Natural Gas	20
PJM West Summer Spark Spreads	22

Introduction

Logical Information Machines (LIM) provides occasional examples of the rich and varied analysis that customers can perform using our patented near-English query tool “XMIM”. The examples included here come from the energy markets in the United States and represent real market opportunities that traders can take advantage of. As you will see, the queries are fairly sophisticated and based on a strong knowledge of the fundamental drivers in the oil and gas markets. We hope that existing LIM users will see value in these queries and that those readers new to LIM will recognize the very high value that our tools add to market data.

LIM is a recognized leader in data management for the energy market as recipient of the 2006 Energy Risk Data Management Award. As well as the fundamental and futures market data you see in these examples, we provide access to over 120 different energy market data vendors and websites.

We hope that you find the examples illuminating and that if you have questions or suggestions about the analysis, you let us know what you think. For more information about XMIM, please contact a LIM sales representative at one of the following offices, or visit our website at www.lim.com. **LIM**

Chicago: +1 (312) 456-3000

Houston: +1 (713) 652-4066

New York: +1 (212) 292-4282

Austin: +1 (512) 697-3000

London: +44 20 7190 2945

Singapore: +(65) 6322 4122

U.S. Petroleum Inventories & NYMEX WTI Price Declines

A concern in the crude oil market on Tuesday is the next day Energy Information Administration (EIA) Petroleum Stocks report. Will petroleum stocks be higher or lower and how will prices react? Traders usually have their own expectations or access to a forecast to help answer the question. Despite the general trend of higher WTI prices since 2001 the one day move of the front CL contract close from Tuesday to Wednesday has been negative 53% of the time. Such a weak trend might not be sufficient to take a short position, but with use of XMIM specific conditions can be isolated and a short position better analyzed.

The query below compares the Wednesday reported inventories to the last 13 weeks. If stocks be at the lowest level, the Tuesday to Wednesday move ("Mov1" column) is shown. The hundreds of possible

Tuesday trades since 2001 are limited to just 67 with a net price move of \$-24.84 p/bbl, for an average of \$-0.37 p/bbl and a 64% short trade success rate. The market move is typically only temporary, on the next day Wednesday to Thursday ("Mov2" column) CL prices rebound. Spot on accurate inventory forecasts should not be an issue with this query. The 13-week trend allows for some inaccuracies and would usually only need the right directional change, either higher or lower weekly inventory level. **LIM**

Query Language

```
LET ATTR CLStk =
  IF weekly_value of DOE.STOCKS.TOTAL.CRUDE is not DEFINED
  THEN weekly_value of DOE.STOCKS.TOTAL.CRUDE on previous
      {weekly_value of DOE.STOCKS.TOTAL.CRUDE is DEFINED}
  ELSE weekly_value of DOE.STOCKS.TOTAL.CRUDE
  ENDIF

SHOW
  CL: Close of CL
  Mov1: 1 value move of close of CL 1 value later
  CLStk: CLStk

WHEN
  Date is from 2001 to 2006
  AND Date is Tuesday
  AND Date is not any holiday
  AND highest from 13 weeks ago to today of CLStk is exactly CLStk
##### 9/11/01
  AND Date is not from 9/07/2001 to 10/11/2001
##### Hurricane Katrina
  AND Date is not from 8/26/2005 to 9/26/2005

OR
  Date is from 2001 to 2006
  AND Date is Wednesday
  AND 1 day ago Date is any holiday
  AND highest from 13 weeks ago to today of CLStk is exactly CLStk

##### 9/11/01
  AND Date is not from 9/07/2001 to 10/11/2001
##### Hurricane Katrina
  AND Date is not from 8/26/2005 to 9/26/2005
```

Query Results

	CL	Mov1	Mov2	CLStk
Sum	2769.6400	-24.8400	3.9700	20776093.0000
Avg	41.3379	-0.3707	0.0593	310090.9403
AvgPos	41.3379	0.6787	0.6344	310090.9403
AvgNeg	NaN	-0.9565	-0.7418	NaN
PctPos	100.0000	35.8209	58.2090	100.0000
PctNeg	0.0000	64.1791	41.7910	0.0000
Maximum	74.6100	1.5500	1.4100	346964.0000
Minimum	18.3400	-3.6400	-2.3400	279509.0000
StdDev	14.7375	1.0619	0.8784	17872.9072
67 Occurrences				

U.S. Petroleum Inventories & NYMEX WTI Prices

NYMEX WTI prices react to the weekly U.S. Energy Information Administration (EIA) Petroleum Stocks reports. Over the last 10 years reported U.S. crude oil stocks have moved from 293 MMbbl to a maximum of 355 MMbbl, down to a minimum of 264 MMbbl and back to the current 293 MMbbl. During this same period on the Thursday following the report the front WTI contract price moved higher 54% of the time and higher 58% of the time by Friday. The net move over the years is \$23 on Thursday and \$61 by Friday. Some of this move is due to changes in inventory levels, some general price advance from \$26 p/bbl in 1997 to the current \$70 p/bbl level and many other reasons.

The XMIM query below isolates particular inventory levels and the weekly inventory change, and identifies 1 and 2-day price moves (Thursday and Friday) from the Wednesday release of the EIA report. Of the almost 500 weekly inventory trade opportunities, special conditions help identify 166 occurrences of more predictable price moves. Specifically, this occurs when petroleum stocks are at least one standard deviation below the 31-week average and the weekly stock change is below -4 MMbbl, or when stocks are one standard deviation above average and the change in stocks is above -4 MMbbl.

The resulting 1 and 2-day moves are an improvement over the 10-year level. The 1-day upward move to Thursday happens 61% of the time and the second day upward move 65%. Summarized from the XMIM output, the success rate improvement was enough to have averaged \$0.31 p/bbl over the 166 trades.

	Day 1 (Thursdays)	Cumulative Day 2 (Fridays)
Low Inventories	\$0.1452	\$0.4345
High Inventories	\$0.1497	\$0.2893
Total:	\$0.1489	\$0.3146

The query is intended to help support taking a CL front contract position as the EIA Petroleum Stocks report is released near the closing of the Wednesday exchange trading. The summary statistics indicate greater price appreciation over two days, enough perhaps to justify another overnight position. The query code also includes a feature to compensate for a Wednesday holiday. **LIM**

Query Language

```

LET ATTR Inven = IF Date is Wednesday
                  AND Date is not any holiday
                  THEN weekly_value of DOE.STOCKS.TOTAL.CRUDE 1 value ago
                  ELSE IF Date is Thursday
                  AND Date is 1 day after any holiday
                  THEN weekly_value of DOE.STOCKS.TOTAL.CRUDE 1 value ago
                  ENDIF
                  ENDIF
LET ATTR Avg     = nearest_integer (31 week average of Inven ,1)
LET ATTR StdDev  = nearest_integer (31 week standard_deviation of Inven ,1 )
LET ATTR InvenSig = IF Inven < (Avg - StdDev )
                    AND 1 week move of Inven < -4000
                    THEN 1
                    ELSE IF Inven >= (Avg + StdDev)
                    AND 1 week move of Inven > -4000
                    THEN 2
                    ENDIF
                    ENDIF
LET ATTR Mov1 = 1 value move of CL 1 value later
LET ATTR Mov2 = 2 value move of CL 2 value later

SHOW
#   Inven: Inven
#   Avg: Avg
#   StdDev: StdDev
#   CL: CL
#   Low_Inven_Day1: IF InvenSig == 1
#                   THEN Mov1
#                   ENDIF
#   Low_Inven_Day2: IF InvenSig == 1
#                   THEN Mov2
#                   ENDIF
#   High_Inven_Day1: IF InvenSig == 2
#                   THEN Mov1
#                   ENDIF
#   High_Inven_Day2: IF InvenSig == 2

```

```

THEN Mov2
ENDIF
WHEN
    Date is from 1997 to 2006
    AND Date is Wednesday
    AND Date is not any holiday
    AND InvenSig > 0
##### 9/11/01
    AND Date is not from 9/11/2001 to 10/11/2001
##### Iraqi War II
    AND Date is not from 3/17/2003 to 4/17/2003
##### Hurricane Ivan
    AND Date is not from 9/13/2004 to 10
    AND Date is not from 8/26/2005 to 9//13/2004
##### Hurricane Katrina    26/2005
OR
    Date is from 1997 to 2006
    AND Date is Thursday
    AND Date is 1 day after any holiday
    AND InvenSig > 0
##### 9/11/01
    AND Date is not from 9/11/2001 to 10/11/2001
##### Iraqi War II
    AND Date is not from 3/17/2003 to 4/17/2003
##### Hurricane Ivan
    AND Date is not from 9/13/2004 to 10/13/2004
##### Hurricane Katrina
    AND Date is not from 8/26/2005 to 9/26/2005
    
```

Query Results

	Low_Inven	Low_Inven	High_Inven	High_Inven
Sum	4.2100	12.6000	20.5100	39.6300
Avg	0.1452	0.4345	0.1497	0.2893
AvgPos	0.4805	0.6239	0.5988	0.8415
AvgNeg	-0.4920	-0.2917	-0.5294	-0.6255
PctPos	65.5172	79.3103	59.8540	62.0438
PctNeg	34.4828	20.6897	39.4164	37.2263
Maximum	1.2300	1.7000	2.1400	3.2800
Minimum	-1.0600	-0.8700	-2.3400	-2.6600
StdDev	0.5610	0.5810	0.7455	0.9695
166 Occurrences				

NYMEX WTI Trend Analysis

Well-established technical trend analysis is a common method for supporting financial and commodity market portfolios. One such technique is to find ascending daily highs as an indication of market strength. While easily described in hindsight, use of trend analysis for the prediction of just a one-day price move can be difficult. Three days of successive increases have occurred 88 times since 2001. Once the three day trend is identified a net price move from day 3 to day 4 of \$11.26 p/bbl has occurred, for an average move of \$0.128 p/bbl and a modest success rate of 61%.

The power of XMIM is well suited for examining past price trends, isolating repeating market characteristics and, prediction of a future daily price move. The query below builds upon the easy to spot rising high price trend by adding daily WTI volumes and statistical evaluations. Volume measurements two days and one day before market entrance must meet specific levels. The daily high prices are

rising, but only near one standard deviation of the daily move. The purpose of the special conditions is to identify modestly rising prices and volumes. Falling volumes or wildly changing prices may signal the end of the trend instead of one more day of higher prices.

Limiting market entrance to these criteria reduces exposure to just 24 trades with nearly the same net price move of \$11.09, for a significantly higher average trade of \$0.462 p/bbl and a higher success rate of 83%. Of note is that market entrance occurs on the same day as an observed high. Should the closing price be near the high quick execution will be required. Also multiple entries into the same trend are disallowed with activation of the First Adjacent Value option and the \$2.61 price move of Hurricane Katrina is excluded. Many permutations of contract volumes and rising prices are possible and you are encouraged to test the limits of the query for yourself. **LIM**

Query Language

```
LET ATTR Hi_Hi2 = 1 value move of High of CL 2 value ago
LET ATTR Hi_Hi1 = 1 value move of High of CL 1 value ago
LET ATTR Hi_Hi   = 1 value move of High of CL

LET ATTR HiAvg  = 31 value average of (1 value move of High of CL)
LET ATTR HiStdv = 31 value standard_deviation of (1 value move of High of CL)

LET ATTR VolTrd  = TotalVol of CL
LET ATTR VolAvg  = 31 value average of TotalVol of CL
LET ATTR VolStdv = 31 value standard_deviation of TotalVol of CL
LET ATTR AvgVolChg = 31 value average of
    1 value move of TotalVol of CL 1 value ago)
LET ATTR VolChgStdv = 31 value standard_deviation of
    1 value move of TotalVol of CL 1 value ago)

SHOW
  CL: Close of CL
  Mov1: 1 value move of CL 1 value later
  VolTrd: VolTrd
  AvgVolChg: AvgVolChg
  StdvVolChg: VolChgStdv
  HiStdv: HiStdv

WHEN
  Date is from 2001 to 2006
  AND 1 value move of TotalVol of CL 2 value ago < (AvgVolChg * 1)
  AND 1 value move of TotalVol of CL 1 value ago > (VolChgStdv * -.5 )
  AND Hi_Hi2 > 0
    AND Hi_Hi2 < (HiStdv * .8 )
  AND Hi_Hi1 > 0
    AND Hi_Hi1 < (HiStdv * 1.0 )
  AND Hi_Hi > 0
    AND Hi_Hi < (HiStdv * 1.25 )

##### 9/11/01
  AND Date is not from 9/07/2001 to 10/11/2001
##### Hurricane Katrina
  AND Date is not from 8/26/2005 to 9/26/2005
```

Query Results

	CL	Mov1	VolTrd	AvgVolChg	StdvVolChg	HiStdv
Sum	868.3100	11.0900	4796567.000	-40675.7097	1135918.2768	15.5476
Avg	36.1796	0.4621	199856.9583	-1694.8212	47329.9282	0.6478
AvgPos	36.1796	0.6305	199856.9583	1882.9534	47329.9282	0.6478
AvgNeg	NaN	-0.3800	NaN	-3841.4860	NaN	NaN
PctPos	10.0000	83.3333	100.0000	37.5000	100.0000	100.0000
PctNeg	0.00000	16.6667	0.0000	62.5000	0.0000	0.0000
Maximum	61.5700	1.6600	289323.0000	4528.9533	62633.2118	1.1109
Minimum	19.9900	-1.0100	121691.0000	-9270.5484	27755.6401	0.3663
StdDev	11.9319	0.6560	46613.4772	3572.5298	8097.1796	0.2165
24 Occurrences						

NYMEX WTI—Commitment of Traders & Small Traders

The front WTI futures contract has high volatility any day of the week. Since 2001 the Tuesday-Wednesday move (close-to-close) has the highest volatility (standard deviation of \$0.947 p/bbl) and the Thursday-Friday move the lowest (standard deviation of \$0.77 p/bbl). The Monday-Tuesday price move up or down maybe the most random since it is the closest to an even 50/50 split with an average move of just \$0.0148 p/bbl. One method to add more predictability to the Monday-Tuesday WTI front contract price move is to use the historical analysis features of XMIM and compare the price moves to the CFTC Commitment of Traders Report (COT) for crude oil futures.

On Monday the COT report is generally available for trading analysis. The “highest” and “lowest” function of XMIM is used to find the extremes of the small trader net position. If the current report

shows either the highest or lowest of the last 13 weeks then a short position could have been opened on the Monday close and exited on the Tuesday close with a better than 50/50 outcome expectation. Upon 60 occasions, the front WTI contract moved down a total of \$-12.49 p/bbl (“1DayMov” column) for an average of \$-0.208 p/bbl and a 70% success rate. The 2DayMov column (Tuesday-Wednesday) also shows an overall drop in prices of \$-8.89 p/bbl, but with a trade success rate of only 51% the extra day may not be advisable. **LIM**

Query Language

```
LET ATTR STTrd = NetSmallTraders(CL) on previous Tuesday
SHOW
    1DayMov: 1 value move of close of CL 1 value later
    2DayMov: 1 value move of close of CL 2 value later
    CL: Close of CL
    SmallTraderNet: STTrd
WHEN
    Date is from 2001 to 2006
    AND Date is Monday
    AND Date is not any holiday
    AND highest from 13 weeks ago to today of STTrd is exactly STTrd
##### 9/11/01
    AND Date is not from 9/07/2001 to 10/11/2001
##### Iraqi War II
    AND Date is not from 3/17/2003 to 4/17/2003
OR
    Date is from 2001 to 2006
    AND Date is Monday
    AND Date is not any holiday
    AND lowest from 13 weeks ago to today of STTrd is exactly STTrd
    AND Date is not from 9/07/2001 to 10/11/2001
    AND Date is not from 3/17/2003 to 4/17/2003
OR
    Date is from 2001 to 2006
    AND Date is Tuesday
    AND Date is 1 day after any holiday
    AND highest from 13 weeks ago to today of STTrd is exactly STTrd
    AND Date is not from 9/07/2001 to 10/11/2001
    AND Date is not from 3/17/2003 to 4/17/2003
OR
    Date is from 2001 to 2006
    AND Date is Tuesday
    AND Date is 1 day after any holiday
    AND lowest from 13 weeks ago to today of STTrd is exactly STTrd
    AND Date is not from 9/07/2001 to 10/11/2001
    AND Date is not from 3/17/2003 to 4/17/2003
```

Query Results

	1DayMov	2DayMov	CL	SmallTraderNet
Sum	-12.4900	-8.8900	2375.8000	-56667.0000
Avg	-0.2082	-0.1482	39.5967	-944.4500
AvgPos	0.6333	0.5755	39.5967	13053.3077
AvgNeg	-0.5688	-0.8252	NaN	-11648.6176
PctPos	30.0000	48.3333	100.0000	43.3333
PctNeg	70.0000	51.6667	0.0000	56.6667
Maximum	2.1200	1.5300	68.7400	32932.0000
Minimum	-2.0200	-2.7100	18.8900	-29708.0000
StdDev	0.7341	0.8831	14.6880	15225.6703

60 Occurrences

NYMEX WTI—Commitment of Traders Graph


Over the last four years the CFTC WTI Commitment of Traders Report shows a growing large speculator long position. The position has change from 12,277 in January 2002 to a recent 175,621 contracts, more than a 10-fold increase. The XMIM generated graph below shows a visual relation between the rise and fall of crude oil prices and the large speculators long position (red and green lines). Use of the correlation

function further confirms the relation. Since 2001 the correlation has greatly increased and was recently near 80% (blue line). **LIM**

Query Language

```
SHOW
  CL: close of CL
  CoTLSL: CoTLSL of CL
  Corr26: 26 week correlation of (CoTLSL of CL / CoTLSL of CL 1 week ago)
  and (close of CL / close of CL 2 week ago)
WHEN
  Date is after 2001
  AND Date is Tuesday
```

Query Results



NYMEX Natural Gas & Inventories

The NYMEX NG summer 2006 contracts have fallen from the \$10.00 p/MMBtu highs of early January to the current \$6-7.00 level. Higher than average natural gas storage and offshore Gulf of Mexico production recovery are much of the reason for the decline. Part of summer price volatility is related to storage activity and storage levels that can be quantified by using the XMIM system.

The query below shows the move from the Wednesday front contract close to the Friday close, two days of activity centered on the EIA Weekly Natural Gas Storage Report. The price moves are further divided between injection activity and overall storage levels. The query has coding for testing any of the summer months. Over the 7-month

injection season the standard deviation of the 2-day price move is about \$0.20 p/MMBtu versus \$0.16 p/MMBtu for the individual month of July displayed. High inventories conditions, represented by either higher than 5-year average injections (“MovHiInj” column) or high storage balances (“MovHiBal” column), tend to support negative price moves averaging \$-0.0278 and \$-0.0253 in July. Lower than average injections or storage balances (“MovLoInj” and “MovLoBal” columns) support average price increases of \$0.0096 and \$0.0246.

The query can be easily modified to show price moves, injection activity and balances during other months. Modify the “WHEN” statement to find a month with high price moves (hint: October). **LIM**

Query Language

```
LET
@NGInv = EIA.TOTAL.US
#
@NGInv = EIA.CONSUMING.E
ATTR Inven5Yr = ((The 1 calendar week GasStorage of @NGInv 1 year ago +
The 1 calendar week GasStorage of @NGInv 2 years ago +
The 1 calendar week GasStorage of @NGInv 3 years ago +
The 1 calendar week GasStorage of @NGInv 4 years ago +
The 1 calendar week GasStorage of @NGInv 5 years ago) / 5)

SHOW
NG: NG
Injection: GasStorageChange of @NGInv
MovHiInj: IF GasStorageChange of @NGInv > (Inven5Yr - Inven5Yr 1 week ago)
THEN 2 value move of close of NG 1 value later
ENDIF
MovLoInj: IF GasStorageChange of @NGInv < (Inven5Yr - Inven5Yr 1 week ago)
THEN 2 value move of close of NG 1 value later
ENDIF
Balance: The 1 calendar week GasStorage of @NGInv
MovHiBal: IF The 1 calendar week GasStorage of @NGInv > Inven5Yr
THEN 2 value move of close of NG 1 value later
ENDIF
MovLoBal: IF The 1 calendar week GasStorage of @NGInv < Inven5Yr
THEN 2 value move of close of NG 1 value later
ENDIF
5YrAvg: Inven5Yr

WHEN
Date is after 1998
#
AND Date is April
#
AND Date is May
#
AND Date is June
AND Date is July
#
AND Date is August
#
AND Date is September
#
AND Date is October
#
AND Date is from April to October
```

Query Results

	NG	Injection	MovHiInj	MovLoInj	Balance	MovHiBal	MovLoBal	5YrAvg
Avg	4.4877	79.1667	-0.0278	0.0096	2103.7333	-0.0253	0.0246	2042.3400
AvgPos	4.4877	79.1667	0.1478	0.1233	2103.7333	0.1347	0.1345	2042.3400
AvgNeg	NaN	NaN	-0.1332	-0.1041	NaN	-0.1054	-0.1953	NaN
PctPos	100.0000	100.0000	37.5000	50.0000	100.0000	33.3333	66.6667	100.0000
PctNeg	0.0000	0.0000	62.5000	50.0000	0.0000	66.6667	33.3333	0.0000
Maximum	7.8440	147.0000	0.2750	0.2930	2486.0000	0.2930	0.2060	2289.2000
Minimum	2.1620	39.0000	-0.3660	-0.2160	1662.0000	-0.2290	-0.3660	1809.4000
StdDev	1.7846	23.6513	0.1702	0.1424	222.1744	0.1430	0.1885	118.1963

30 Occurrences

NYMEX Natural Gas & Summer Temperatures

Warm summer temperatures in the Northeast increase the use of electricity for space cooling that in turn increases demand for natural gas fired power generation. Between the 2nd and 3rd quarters an extra 6 billion cubic feet of gas per day is consumed by the power industry and average historical volatility increases by 5%. The seasonal demand increase is built into future contract prices but the added volatility may not. Some of the higher volatility is due to market price swings caused by the above normal temperatures of a “heat wave.” The XMIM query below searches the LIM database for short term temperature patterns and tracks the natural gas prices in advance and during a “heat wave” event. Future contract prices respond to weather expectations ranging several days to weeks in advance and are crucial for capturing the price volatility above normal temperatures create.

Important details of the query include the city selection, the temperature and the forecast period. Average temperature measurements at Newark NJ and Philadelphia PA generated slightly better results than other major Eastern cities and are a part of the query. The query is designed to isolate three day weather patterns with specific characteristics, day 3, 4 and day 5 must be “X” degrees above normal. A wide range of values (-10 to 14) were substituted for X degrees and tested. The lower range did not result in predictable price movements and the higher ranges occurred infrequently. The most successful tested value was 6 degrees.

The look-back nature of this query substitutes the actual temperature for the weatherman’s forecast, creating a perfect forecast. Highly accurate, short term, weather predictions allow a

forecast to be a workable substitute for the look-back of the query. Accuracy decays with time which places a reasonable limit of five days on this query. Should the client’s LIM installation not include weather forecasts, the broadcast media or internet website forecasts should be adequate for the 3 to 5-day predictions.

The results of the query, shown in the Mov3 column below, indicate a successful trade pattern averaging \$0.1543 p/MMBtu. Over the eight past years the success rate is 79% on 39 trades. Depending on temperatures, trading activity is higher in some years than others. The query produces 13 trades for the year 2001, while 2004 produces none at all. Only 1998 generates a loss, all other years show earnings.

It should be noted that the query excludes many days during the USGC hurricane season. Hurricanes are major supply disruptions, not temperature driven demand events and beyond the scope of this query. Should the heat wave extend many days, more than one position will be open. Following the XMIM summary statistics is a second query to establish the normal based temperature constraints the weatherman’s forecast must exceed.

For example, on the first day of output below (Thursday, 08/20/1998), a front NG contract entry signal occurs if the temperature forecasts for Sunday, Monday and Tuesday are higher than 84, 90 and 90 respectively. On Tuesday, three trading days later, market reaction to the “heat wave” has passed and the position is closed for a \$0.092 per MMBtu loss. **LIM**

Query Language

```
##### Use an average of two cities representing much of the
##### Eastern U.S. population and gas demand
LET
  ATTR AvgHigh = nearest_integer
  ((HighTemp of NEWARK.NJ +
  HighTemp of PHILADELPHIA.PA) * .5, 1)
LET
  ATTR AvgNrmHigh = nearest_integer
  ((NormalHighTemp of NEWARK.NJ +
  NormalHighTemp of PHILADELPHIA.PA) * .5, 1)

##### The degrees above normal related to predictable price moves
LET ATTR DegAdder = 6
SHOW
  NG: Close of NG
##### Track the price changes of the natural gas contract
Mov0: Close of NG - Open of NG
Mov1: Close of NG 1 value later - Open of NG
Mov2: Close of NG 2 value later - Open of NG
Mov3: Close of NG 3 value later - Open of NG

##### Forecast temperatures from 3 to 5 days out
##### should be lower or higher than these values
Fst3>: (AvgNrmHigh 3 day later + (DegAdder * 0))
Fst4>: (AvgNrmHigh 4 day later + (DegAdder * 1))
Fst5>: (AvgNrmHigh 5 day later + (DegAdder * 1))
```

```

WHEN
    Date is from 7/1/1998 to 9/30/2006
    AND Date is third quarter

##### Select sequential days when temperatures are near normal or
##### increasing to above normal
    AND AvgHigh 3 day later > (AvgNrmHigh 3 day later + (DegAdder * 0))
    AND AvgHigh 4 day later > (AvgNrmHigh 4 day later + (DegAdder * 1))
    AND AvgHigh 5 day later > (AvgNrmHigh 5 day later + (DegAdder * 1))

    AND Date is not Saturday
    AND Date is not Sunday
    AND Date is not any holiday
    AND Date is not any semi_holiday

##### Remove production impact from Hurricanes
    AND Date is not from 8/27/2005 to 12/31/2005 ### Katrina
    AND Date is not from 7/7/2005 to 7/16/2005 ### Dennis
    AND Date is not from 9/13/2004 to 12/31/2004 ### Ivan
    AND Date is not from 8/11/2004 to 8/13/2004 ### Charley
    AND Date is not from 9/23/2002 to 9/28/2002 ### Isidore
    AND Date is not from 9/9/1998 to 9/12/1998 ### Frances

##### Code for weather forecast constraints
##### Use an average of two cities representing much of the
##### Eastern U.S. population and gas demand
LET
    ATTR AvgHigh = nearest_integer
        ((HighTemp of NEWARK.NJ +
         HighTemp of PHILADELPHIA.PA) * .5, 1)
LET
    ATTR AvgNrmHigh = nearest_integer
        ((NormalHighTemp of NEWARK.NJ +
         NormalHighTemp of PHILADELPHIA.PA) * .5, 1)
##### The degrees above normal related to predictable price moves

LET ATTR DegAdder = 6

SHOW
##### Forecast should be below or above these normal based temperatures
    Deg3>: (AvgNrmHigh 3 day later + (DegAdder * 0))
    Deg4>: (AvgNrmHigh 4 day later + (DegAdder * 1))
    Deg5>: (AvgNrmHigh 5 day later + (DegAdder * 1))

WHEN
    Date is 2006
    AND Date is third quarter
    
```

Query Results

Date	Day	NG	Mov0	Mov1	Mov2	Mov3	Fst3>	Fst4>	Fst5>
08/20/1998	Thu	1.9530	0.0330	0.0270	0.0060	-0.0920	84.0000	90.0000	90.0000
	Sum	142.7820	0.1570	1.3370	3.5700	6.0170	3250.0000	3478.0000	3473.000
	Avg	3.661	0.0040	0.0343	0.0915	0.1543	83.3333	89.1795	89.0513
	AvgPos	3.661	0.0716	0.1225	0.1755	0.2380	83.3333	89.1795	89.0513
	AvgNeg	NaN	-0.0748	-0.1422	-0.1883	-0.1701	NaN	NaN	NaN
	PctPos	100.0000	53.8462	66.6667	76.9231	79.4872	100.0000	100.0000	100.0000
	PctNeg	0.0000	46.1538	33.3333	23.0769	20.5128	0.0000	0.0000	0.0000
	Maximum	9.3910	0.2290	0.4810	0.7110	0.9980	88.0000	94.0000	94.0000
	Minimum	1.6520	-0.4240	-0.8870	-0.7040	-0.2800	72.0000	77.0000	77.0000
	StdDev	2.0668	0.1064	0.2064	0.2321	0.2507	3.8752	4.0902	4.0649

39 Occurrences

NYMEX Natural Gas Prices & Hurricanes

The 2006 hurricane season brings the threat of new Gulf of Mexico production disruptions. The uncertainty about the ultimate storm wind speed and direction is an important determinate of price volatility as is any actual damage. The term of disruption is most often just a few days as wells are shut-in and crews removed from the production platforms. The impact can be also be devastating as with Hurricane Katrina which nine months later was still responsible for more than 300,000 bbl/d of oil and one Bcf/d of natural gas shut-ins. Over the last 10 years, 12 hurricane-strength storms (shown following) have made U.S. Gulf States production area landfall and had actual or at least a threat of supply impact.

The query below displays the daily price change as the storms move through the Gulf of Mexico from five days before landfall right up to crossing the coastline. Across all 12 storms price increases occur several days in advance of landfall as the futures market anticipates a supply disruption. The XMIM statistical summary shows that an average price increase of over \$0.05 p/MMBtu with success rate of 66% occurs three and five days before landfall. The lowest tested success rate of 33% is on the day of landfall as the storm and volatility diminish. Note:

two hurricanes, Cindy and Dennis have been added to the standard XMIM files and several tropical storms, such as Bertha in 2002 are omitted due to achieving only sub-hurricane wind speeds. **LIM**

Year	Storm	Gulf Landfall
1997	Danny	LA., July 19
1998	Earl	FL., September 2
1998	Georges	MS., September 28
1999	Bret	TX., August 22
2002	Lilly	LA., October 3
2003	Claudette	TX., July 15
2004	Ivan	AL., September 16
2005	Cindy	LA., July 6
2005	Dennis	FL., July 10
2005	Emily	TX., July 20
2005	Katrina	LA., August 29
2005	Rita	TX., September 24

Query Language

```
SHOW
  NG: close of NG
  Mov: 1 day move of close of NG
  Mov-1: 1 day move of close of NG 1 days ago
  Mov-2: 1 day move of close of NG 2 days ago
  Mov-3: 1 day move of close of NG 3 days ago
  Mov-4: 1 day move of close of NG 4 days ago
  Mov-5: 1 day move of close of NG 5 days ago
  Date: TransDate of DATE.USA_HURRICANES
  # 1: Landfall of DATE.USA_HURRICANES_TEXAS
  # 2: Landfall of DATE.USA_HURRICANES_LOUISIANA
  # 3: Landfall of DATE.USA_HURRICANES_MISSISSIPPI
  # 4: Landfall of DATE.USA_HURRICANES_ALABAMA
  # 5: Landfall of DATE.USA_HURRICANES_FLORIDA

WHEN
  Date is after 1996
    AND Landfall of DATE.USA_HURRICANES_TEXAS is DEFINED
  OR Date is after 1996
    AND Landfall of DATE.USA_HURRICANES_LOUISIANA is DEFINED
    AND Date is not 9/26/2005
  OR Date is after 1996
    AND Landfall of DATE.USA_HURRICANES_MISSISSIPPI is DEFINED
  OR Date is after 1996
    AND Landfall of DATE.USA_HURRICANES_ALABAMA is DEFINED
##### FL Dennis
  OR Date is 7/10/2005
##### LA Cindy
  OR Date is 7/6/2005
```

Query Results

	NG	Mov	Mov-1	Mov-2	Mov-3	Mov-4	Mov-5	Date
Avg	5.4039	-0.0028	0.0205	0.0033	0.0547	0.1232	0.0569	20022491.1667
AvgPos	5.4039	0.3455	0.0947	0.0893	0.1495	0.3728	0.1294	20022491.1667
AvgNeg	NaN	-0.1770	-0.1280	-0.1465	-0.1350	-0.0643	-0.0880	NaN
PctPos	100.0000	33.3333	66.6667	58.3333	66.6667	41.6667	66.6667	100.0000
PctNeg	0.0000	66.6667	33.3333	33.3333	33.3333	50.0000	33.3333	0.0000
Maximum	10.8470	1.0550	0.3040	0.2130	0.3040	1.5190	0.4530	20050924.0000
Minimum	1.6520	-0.4660	-0.2900	-0.2140	-0.2620	-0.1060	-0.1920	19970719.0000
StdDev	3.1246	0.3852	0.1505	0.1277	0.1846	0.4470	0.1659	32404.3694

12 Occurrences

Winter Season NYMEX Natural Gas

Heating Degree Days (HDDs) and natural gas storage levels are among the most important influences on winter natural gas prices. Extremes such as the 67 HDDs (-2 degrees) for the Boston/Baltimore low in January 1994 or the inventory withdrawals of more than 250 Bcf in January 1997 and February 2000 can move NYMEX prices.

The XMIM query below compares current inventory levels to historical averages, and season-to-date HDD to normal weather. Historically, when inventories are at normal or lower levels and season-to-date HDD are slightly below normal or higher (colder), NG prices on

average increase for the next three days. Over the past several years when such conditions are met, the Day3 prices moved up an average of \$0.40 p/MMBtu and increased more than 75% of the time. (Note: set the XMIM Options Execute Display/Sort to First Value in 4 Days.) **LIM**

Query Language

```

LET
  ATTR AvgLow = (LowTemp of BOSTON.LOGAN.MA + LowTemp of BALTIMORE.WASHINGTON.MD) / 2
  ATTR AvgNrmLow = (NormalLowTemp of BOSTON.LOGAN.MA + NormalLowTemp of BALTIMORE.WASHINGTON.MD) / 2
LET
  ATTR HDDSprd = sum from 10/29/_ to today of
    IF AvgLow is less than 65
      THEN nearest_integer (65 - AvgLow, 1)
    ELSE 0
    ENDIF - sum from 10/29/_ to today of
  IF AvgNrmLow is less than 65
    THEN nearest_integer (65 - AvgNrmLow, 1)
    ELSE 0
  ENDIF
LET
  ATTR Inven5Yr = nearest_integer (
    ((The 1 calendar week GasStorage of EIA.TOTAL.US 1 year ago +
      The 1 calendar week GasStorage of EIA.TOTAL.US 2 years ago +
      The 1 calendar week GasStorage of EIA.TOTAL.US 3 years ago +
      The 1 calendar week GasStorage of EIA.TOTAL.US 4 years ago +
      The 1 calendar week GasStorage of EIA.TOTAL.US 5 years ago) / 5), 1)
SHOW
  NG: Close of NG
  InvenSprd: The 1 calendar week GasStorage of EIA.TOTAL.US - Inven5Yr
  HDDSprd: IF Date is from 10/29/_ to 2/23/_
    THEN HDDSprd
    ENDIF
  Day1: IF Date is from 10/29/_ to 2/23/_
    AND HDDSprd is more than -150
    AND HDDSprd is more than HDDSprd 1 week ago
    THEN move from today to 1 value later of Close of NG
    ENDIF
  Day2: IF Date is from 10/29/_ to 2/23/_
    AND HDDSprd is more than -150
    AND HDDSprd is more than HDDSprd 1 week ago
    THEN move from today to 2 value later of Close of NG
    ENDIF
  Day3: IF Date is from 10/29/_ to 2/23/_
    AND HDDSprd is more than -150
    AND HDDSprd is more than HDDSprd 1 week ago
    THEN move from today to 3 value later of Close of NG
    ENDIF
WHEN
  Date is from 4/1/1991 to 2/23/2006
  AND Date is not any holiday
  AND Date is not any semi_holiday
  AND Date is not Saturday
  AND Date is not Sunday
  AND the 1 calendar week GasStorage of EIA.TOTAL.US is less than
    (Inven5Yr + 0)

```

Query Results

	NG	InvenSprd	HDDSprd	Day1	Day2	Day3
Sum	814.0950	-39088.0000	3928.0000	1.2790	3.6210	8.8460
Avg	5.1525	-244.3000	87.2889	0.0581	0.1646	0.4021
AvgPos	5.1525	NaN	128.3243	0.1556	0.2993	0.5981
AvgNeg	NaN	-244.3000	-102.5000	-0.2018	-0.2934	-0.2644
PctPos	100.0000	0.0000	82.2222	72.7273	77.2727	77.2727
PctNeg	0.0000	100.0000	17.7778	27.2727	22.7273	22.7273
Maximum	9.8050	-3.0000	256.0000	0.5750	0.8440	3.0030
Minimum	2.5300	-711.0000	-136.0000	-0.6080	-0.7320	-0.9810
StdDev	1.3230	187.4420	114.0888	0.2242	0.3653	0.7539
586 Occurrences						

PJM West Summer Spark Spreads

The Northeast summer cooling season represents electricity generation opportunities for gas-fired power plants that might not operate during the other months of lower electricity demand. In the Pennsylvania New Jersey Maryland Interconnection (PJM) the relation between natural gas prices, plant efficiency and PJM West summer prices are often favorable for plant operation. Since 2002, the peak day generic spark spread for a 7,000 heat rate unit has averaged \$14.80 per megawatt (p/MW) and \$24.80 p/MW in 2005. As the cost of generation, gas prices or the heat rate rise, the spark spread decreases and as power prices rise, so does the spark.

The query below uses LIM stored temperature, PJM hourly power load, and day-ahead natural gas and electricity market price variables from the Platt's division of McGraw Hill. The multiple outputs show

how above normal temperatures increase power demand which typically supports a higher spark spread (power price - (natural gas price x 7)). In this case 7,000 heat rate economics are used. The query is designed to accept different heat rates and show the lower spark spread less efficient units might earn. For instance, the two degrees above normal case below shows an average spark spread of \$18.93 p/MW and an average peak hour load of 46,904 megawatts over 113 observations compared to a much higher spark and load of \$32.16 p/MW and 49,586 MW when the temperature is eight degrees above normal. Many other costs are important in power plant operations such as local gas distribution, plant startup and transmission costs but the basic spark spread calculation is still a useful indicator of plant economic performance. **LIM**

Query Language

```
LET ATTR Deg = HighTemp of PHILADELPHIA.PA
LET ATTR NrmDeg = NormalHighTemp of PHILADELPHIA.PA

LET ATTR PJMPeak = 1 calendar day highest of The 1 hour Load of
                    PJM.PRELIM.LOAD.PJM

LET ATTR HeatRate = 7
##### PJMW Peak Load & TexasEastern M-3 Prices
LET ATTR Spark7 = Index of WEBDA20 - (Index of IGBEL21 * HeatRate)

LET ATTR AboveNrm = 2 TO 12 BY 2

SHOW
  Deg:      IF Deg >= (NrmDeg + AboveNrm ) THEN Deg ENDIF
  $M-3:     IF Deg >= (NrmDeg + AboveNrm ) THEN Index of IGBEL21 ENDIF
  $PJM:     IF Deg >= (NrmDeg + AboveNrm ) THEN Index of WEBDA20 ENDIF
  $Spark:   IF Deg >= (NrmDeg + AboveNrm ) THEN Spark7 ENDIF
  PJMPeak: IF Deg >= (NrmDeg + AboveNrm ) THEN PJMPeak ENDIF

WHEN
  Date is after 2002
  AND Date is from Monday to Friday
  AND Date is from July to September
  AND Deg >= (NrmDeg + AboveNrm )
```

Query Results

	Deg	\$M-3	\$PJM	\$Spark	PJMPeak
HeatRate =		7			
AboveNrm =		2.000000			
Avg	87.1947	7.8842	74.1330	18.9316	46904.3452
AvgPos	87.1947	7.8842	74.1330	19.6712	46904.3452
AvgNeg	NaN	NaN	NaN	-7.2000	NaN
PctPos	100.0000	100.0000	100.0000	97.2477	100.0000
PctNeg	0.0000	0.0000	0.0000	2.7523	0.0000
Maximum	98.0000	16.5950	153.3300	85.1850	59500.3984
Minimum	75.0000	4.6000	33.7400	-14.6200	34845.6992
StdDev	4.8658	3.1869	31.0825	17.1401	6249.3808

113 Occurrences

		Deg	\$M-3	\$PJMW	\$Spark	PJMPeak
HeatRate	=	7				
AboveNrm	=	4.000000				
Avg		88.0602	8.2460	80.0189	21.8675	47600.1197
AvgPos		88.0602	8.2460	80.0189	22.5094	47600.1197
AvgNeg		NaN	NaN	NaN	-3.4900	NaN
PctPos		100.0000	100.0000	100.0000	97.5309	100.0000
PctNeg		0.0000	0.0000	0.0000	2.4691	0.0000
Maximum		98.0000	16.5950	153.3300	85.1850	59500.3984
Minimum		77.0000	4.6000	33.7400	-6.7000	34845.6992
StdDev		4.9098	3.3314	32.5536	18.5508	6448.4346

83 Occurrences

		Deg	\$M-3	\$PJMW	\$Spark	PJMPeak
HeatRate	=	7				
AboveNrm	=	6.000000				
Avg		89.2456	8.4634	85.0665	25.1345	48784.9000
AvgPos		89.2456	8.24634	85.0665	26.2146	48784.9000
AvgNeg		NaN	NaN	NaN	-3.4900	NaN
PctPos		100.0000	100.0000	100.0000	96.3636	100.0000
PctNeg		0.0000	0.0000	0.0000	3.6364	0.0000
Maximum		98.0000	16.5950	153.3300	85.1850	59500.3984
Minimum		78.0000	4.6000	33.7400	-6.7000	34845.6992
StdDev		4.8817	3.3936	32.9855	19.9947	6607.4466

57 Occurrences

		Deg	\$M-3	\$PJMW	\$Spark	PJMPeak
HeatRate	=	7				
AboveNrm	=	8.000000				
Avg		90.5833	10.3340	107.4650	32.2166	49586.3390
AvgPos		90.5833	10.3340	107.4650	32.2166	49586.3390
AvgNeg		NaN	NaN	NaN	NaN	NaN
PctPos		100.0000	100.0000	100.0000	100.0000	100.0000
PctNeg		0.0000	0.0000	0.0000	0.0000	0.0000
Maximum		98.0000	16.5950	153.3300	85.1850	59500.3984
Minimum		83.0000	5.0100	42.8700	7.7500	38810.8008
StdDev		4.8087	3.6457	29.7268	22.6510	6580.8681

24 Occurrences

HeatRate = 7		AboveNrm = 10.000000			
	Deg	\$M-3	\$PJMW	\$Spark	PJMPeak
Avg	91.0526	10.5255	108.5378	32.8231	49933.6419
AvgPos	91.0526	10.5255	108.5378	32.8231	49933.6419
AvgNeg	NaN	NaN	NaN	NaN	NaN
PctPos	100.0000	100.0000	100.0000	100.0000	100.0000
PctNeg	0.0000	0.0000	0.0000	0.0000	0.0000
Maximum	98.0000	16.5950	153.3300	85.1850	59500.3984
Minimum	83.0000	5.0100	42.8700	7.7500	40883.6992
StdDev	4.7078	3.6865	29.0481	24.3344	6506.8931

19 Occurrences

HeatRate = 7		AboveNrm = 12.000000			
	Deg	\$M-3	\$PJMW	\$Spark	PJMPeak
Avg	92.0000	10.7425	105.9067	30.7092	50821.9661
AvgPos	92.0000	10.7425	105.9067	30.7092	50821.9661
AvgNeg	NaN	NaN	NaN	NaN	NaN
PctPos	100.0000	100.0000	100.0000	100.0000	100.0000
PctNeg	0.0000	0.0000	0.0000	0.0000	0.0000
Maximum	98.0000	16.2850	140.1500	58.5250	59500.3984
Minimum	86.0000	5.9550	52.6700	7.7500	41666.8984
StdDev	4.6904	3.5504	30.0802	20.5696	6805.8668

6 Occurrences